

The Dottie Number

Lawrence C. Paulson

16 June 2026

Abstract

The Dottie number is the unique fixed point d of the cosine function: $\cos d = d$. It is approximately 0.739085133215 and has no known closed form.

This theory establishes the Dottie number's key properties: the fixed point exists (by the intermediate value theorem) and is unique (because $\cos x - x$ has a strictly negative derivative). Next, the value of d to 12 decimal places is shown using the **approximation** proof method. Two more properties of d are also shown: first, that it is *transcendental* (via the Hermite–Lindemann–Weierstrass theorem); second, that it is a *universal attractor*, in the sense that iterating the cosine function from any real starting point converges to it.

Contents

1	The Dottie Number	2
1.1	Existence	2
1.2	Uniqueness	3
1.3	Approximation	4
1.4	The Dottie number is a universal attractor	5
1.5	A trigonometric identity	7
1.6	Transcendence	8

1 The Dottie Number

```
theory Dottie
imports HOL-Analysis.Analysis
          HOL-Decision-Procs.Approximation
          Hermite-Lindemann.Hermite-Lindemann
```

```
begin
```

The Dottie number, approximately 0.739085133215, is the unique fixed point of the cosine function.

This theory establishes the Dottie number's basic theory. We first show that the fixed point *exists* (by the intermediate value theorem) and is *unique* (because $\cos x - x$ has a strictly negative derivative), justifying the definition of *dottie*, and we pin it down to twelve decimal places. We then prove three further results: that the Dottie number is a *universal attractor*, in the sense that iterating cosine from any real starting point converges to it; the trigonometric identity $\sin(\text{dottie}) = \sqrt{1 - \text{dottie}^2}$; and, using the Hermite–Lindemann–Weierstraß theorem, that the Dottie number is *transcendental*.

```
definition dottie :: real where
  dottie  $\equiv$  THE  $x$ .  $\cos x = x$ 
```

```
lemma cos-1-lt-1:  $\cos (1::real) < 1$ 
using cos-monotone-0-pi pi-gt3 by force
```

We shall reason about the function $g(x) = \cos x - x$. The locale provides a scope for g and its properties, which are used by several of the lemmas below.

```
locale Dottie =
  fixes  $g :: real \Rightarrow real$ 
  defines  $g \equiv \lambda x::real. \cos x - x$ 
```

```
begin
```

```
lemma g-has-negative-deriv:
  assumes  $|t| \leq 1$ 
  shows  $\exists d. (g \text{ has-real-derivative } d) (at\ t) \wedge d < 0$ 
proof (intro exI conjI)
  show  $(g \text{ has-real-derivative } (- \sin t - 1)) (at\ t)$ 
    unfolding g-def by (auto intro!: derivative-eq-intros)
  show  $-\sin t - 1 < 0$ 
    using assms pi-gt3 le-arcsin-iff [of - t] by fastforce
qed
```

1.1 Existence

We have $g(0) = 1 > 0$ and $g(1) = \cos 1 - 1 < 0$. Since g is continuous, the intermediate value theorem gives a point $x \in (0, 1)$ where $g(x) = 0$, i.e.

$\cos x = x$.

lemma *dottie-exists*: $\exists x::\text{real}. 0 < x \wedge x < 1 \wedge \cos x = x$

proof –

— Apply the IVT to g on the unit interval at 0.

have *g-cont*: *continuous-on* $\{0..1\}$ g

unfolding *g-def* **by** (*intro continuous-intros*)

obtain $g\ 0 = 1\ g\ 1 < 0$ **using** *cos-1-lt-1* **by** (*simp add: g-def*)

with *IVT2*[*of g 1 0 0*] *g-cont*

obtain x **where** *hx*: $0 \leq x\ x \leq 1\ g\ x = 0$

by (*metis less-eq-real-def zero-le-one*)

hence *cos-eq*: $\cos x = x$ **by** (*simp add: g-def*)

with *hx* **show** *?thesis*

by (*metis cos-1-lt-1 cos-zero order-less-le*)

qed

1.2 Uniqueness

The function $g(x) = \cos x - x$ has derivative $g'(x) = -\sin x - 1$, which is strictly negative for $x \in [-1, 1]$ (since $\sin x \geq 0$ there). A function with strictly negative derivative is strictly decreasing, so g can have at most one zero. We can extend uniqueness to the entire real line.

lemma *dottie-unique*:

fixes $x\ y :: \text{real}$

assumes $\cos x = x\ \cos y = y$

shows $x = y$

proof (*rule ccontr*)

assume $x \neq y$

have *gx*: $g\ x = 0$ **and** *gy*: $g\ y = 0$ **using** *assms* **by** (*auto simp: g-def*)

— The derivative of g is $\lambda x. -\sin x - 1$, which is negative on $\{-1..1\}$.

show *False*

proof (*cases* $|x| > 1 \vee |y| > 1$)

case *True*

then show *?thesis*

by (*metis assms abs-cos-le-one not-less*)

next

case *False*

then have $|x| \leq 1 \wedge |y| \leq 1$

by *simp*

moreover have $x < y \vee y < x$ **using** $\langle x \neq y \rangle$ **by** *linarith*

ultimately show *?thesis*

using *DERIV-neg-imp-decreasing* [*OF - g-has-negative-deriv*] *gx gy*

by *force*

qed

qed

lemma *facts*: $0 < \text{dottie}\ \text{dottie} < 1\ \cos\ \text{dottie} = \text{dottie}$

proof –

obtain $x :: \text{real}$ **where** *hx*: $0 < x\ x < 1\ \cos x = x$

```

    using dottie-exists by blast
  have unique:  $y = x$  if  $\cos y = y$  for  $y :: \text{real}$ 
    by (simp add: dottie-unique  $\langle \cos x = x \rangle$  that)
  have the-eq:  $dottie = x$ 
    unfolding dottie-def using  $\langle \cos x = x \rangle$  unique by blast
  then show  $0 < dottie$   $dottie < 1$   $\cos dottie = dottie$ 
    using hx by (auto simp: g-def)
qed

```

1.3 Approximation

We pin down the Dottie number to 12 decimal places. Note that g is decreasing. We check that $\cos(lb) > lb$ (so the fixed point is above lb) and $\cos(ub) < u$ (so it is below ub).

```

definition lb::real where  $lb \equiv 0.739085133215$ 

```

```

definition ub::real where  $ub \equiv 0.739085133216$ 

```

```

lemma lb-gt:  $\cos lb > lb$ 
  unfolding lb-def
  by (approximation 50)

```

```

lemma ub-lt:  $\cos ub < ub$ 
  unfolding ub-def
  by (approximation 50)

```

```

lemma lb:  $lb < dottie$ 
proof (rule ccontr)
  assume neg:  $\neg lb < dottie$ 
  have gd:  $g lb > 0$ 
    using facts lb-gt by (auto simp: g-def)
  show False
    using DERIV-neg-imp-decreasing [OF - g-has-negative-deriv] facts neg
    by (smt (verit, ccfv-SIG) cos-le-one cos-monotone-0-pi lb-gt pi-gt3)
qed

```

```

lemma ub:  $ub > dottie$ 
proof (rule ccontr)
  assume neg:  $\neg ub > dottie$ 
  have gd:  $g ub < 0$ 
    using facts ub-lt by (auto simp: g-def)
  show False
    using DERIV-neg-imp-decreasing [OF - g-has-negative-deriv] facts neg
    by (smt (verit) cos-ge-minus-one ub-lt gd g-def)
qed

```

1.4 The Dottie number is a universal attractor

Iterating cosine from *any* real starting point converges to the Dottie number. The key fact is that \cos is a contraction on $[-1, 1]$ with Lipschitz constant $\sin 1 < 1$ (since $|\cos' x| = |\sin x| \leq \sin 1$ there), and that \cos maps all of \mathbb{R} into $[-1, 1]$.

lemma *sin1-bounds*: $0 < \sin (1::real) \sin (1::real) < 1$

proof –

have $lt: (1::real) < \pi$ **using** *pi-gt3* **by** *simp*

have $0 < (1::real)$ **by** *simp*

from *sin-gt-zero[OF this lt]* **show** $0 < \sin (1::real)$.

next

have $\sin 1 < \sin (\pi/2)$

using *sin-monotone-2pi[of 1 pi/2]* *pi-gt3* **by** *simp*

then show $\sin (1::real) < 1$ **by** *simp*

qed

lemma *abs-sin-le-sin1*:

assumes $|t| \leq 1$ **shows** $|\sin t| \leq \sin (1::real)$

proof –

have $1 < \pi/2$ **using** *pi-gt3* **by** *simp*

then show *?thesis*

by (*smt (verit, best) assms sin-minus sin-monotone-2pi-le*)

qed

The mean value theorem turns the derivative bound into a Lipschitz bound.

lemma *cos-contraction-lt*:

fixes $x y :: real$

assumes $x < y \ |x| \leq 1 \ |y| \leq 1$

shows $|\cos x - \cos y| \leq \sin 1 * |x - y|$

proof –

have *cont: continuous-on {x..y} cos* **by** (*intro continuous-intros*)

have *deriv: ((cos::real \Rightarrow real) has-derivative (*) (- sin u)) (at u)* **for** $u :: real$

using *DERIV-cos[of u]* **unfolding** *has-field-derivative-def* **by** *simp*

have $\exists \xi \in \{x <..<y\}. \text{norm} (\cos y - \cos x) \leq \text{norm} ((*) (- \sin \xi) (y - x))$

by (*rule mvt-general[OF <x < y> cont]*) (*use deriv in blast*)

then obtain ξ **where** $\xi: \xi \in \{x <..<y\} \ \text{norm} (\cos y - \cos x) \leq \text{norm} (- \sin \xi$

$* (y - x))$

by *auto*

have $|\xi| \leq 1$ **using** ξ *assms* **by** *auto*

then have *absxi: |sin ξ | $\leq \sin 1$* **by** (*rule abs-sin-le-sin1*)

have $|\cos y - \cos x| \leq |\sin \xi| * |y - x|$

using $\xi(2)$ **by** (*simp add: abs-mult*)

also have $\dots \leq \sin 1 * |y - x|$

using *absxi* **by** (*simp add: mult-right-mono*)

finally show *?thesis*

by (*simp add: abs-minus-commute*)

qed

lemma *cos-contraction*:
fixes $x\ y :: \text{real}$
assumes $|x| \leq 1\ |y| \leq 1$
shows $|\cos x - \cos y| \leq \sin 1 * |x - y|$
using *cos-contraction-lt*[of $x\ y$] *cos-contraction-lt*[of $y\ x$] *assms*
by (*cases* $x\ y$ *rule*: *linorder-cases*) (*auto simp*: *abs-minus-commute*)

lemma *dottie-in-pm1*: $|dottie| \leq 1$
using *facts* **by** *simp*

lemma *cos-step-to-dottie*:
assumes $|w| \leq 1$
shows $|\cos w - dottie| \leq \sin 1 * |w - dottie|$
using *facts* **by** (*metis* *assms* *cos-contraction* *dottie-in-pm1*)

After one step the iteration lands in $[-1, 1]$ and stays there.

lemma *cos-funpow-in-pm1*:
fixes $x0 :: \text{real}$
assumes $n \geq 1$
shows $|(\cos \wedge\wedge n)\ x0| \leq 1$
proof –
obtain m **where** $n = \text{Suc } m$ **using** *assms* *not0-implies-Suc* **by** *force*
then **have** $(\cos \wedge\wedge n)\ x0 = \cos ((\cos \wedge\wedge m)\ x0)$
by (*simp* *add*: *funpow-swap1*)
then **show** *?thesis* **by** (*simp* *add*: *abs-cos-le-one*)
qed

From a start in $[-1, 1]$, the distance to the fixed point decays geometrically.

lemma *cos-funpow-bound*:
fixes $y0 :: \text{real}$
assumes $|y0| \leq 1$
shows $|(\cos \wedge\wedge n)\ y0 - dottie| \leq (\sin 1) \wedge n * |y0 - dottie|$
proof (*induction* n)
case 0
show *?case* **by** *simp*
next
case ($\text{Suc } n$)
have *inpm1*: $|(\cos \wedge\wedge n)\ y0| \leq 1$
using *cos-funpow-in-pm1*[of $n\ y0$] *assms*
by (*metis* *funpow-0* *less-one* *not-le*)
have $|(\cos \wedge\wedge \text{Suc } n)\ y0 - dottie| = |\cos ((\cos \wedge\wedge n)\ y0) - dottie|$
by (*simp* *add*: *funpow-swap1*)
also **have** $\dots \leq \sin 1 * |(\cos \wedge\wedge n)\ y0 - dottie|$
using *cos-step-to-dottie*[OF *inpm1*].
also **have** $\dots \leq \sin 1 * ((\sin 1) \wedge n * |y0 - dottie|)$
using *Suc.IH* *sin1-bounds*(1) **by** (*simp* *add*: *mult-left-mono*)
also **have** $\dots = (\sin 1) \wedge (\text{Suc } n) * |y0 - dottie|$

by (*simp add: mult.assoc*)
 finally show ?case .
 qed

lemma *cos-iter-tendsto-unit*:

fixes $y0 :: real$
 assumes $|y0| \leq 1$
 shows $(\lambda n. (\cos \wedge \wedge n) y0) \longrightarrow dotted$
proof –
 have $pow0: (\lambda n. (\sin 1) \wedge n) \longrightarrow (0::real)$
 using *sin1-bounds* by (*intro LIMSEQ-realpow-zero*) *auto*
 have $null: (\lambda n. (\sin 1) \wedge n * |y0 - dotted|) \longrightarrow 0$
 using *tendsto-mult*[*OF pow0 tendsto-const, of |y0 - dotted|*] by *simp*
 have $(\lambda n. |(\cos \wedge \wedge n) y0 - dotted|) \longrightarrow 0$
 using *tendsto-sandwich*[*OF - - tendsto-const null*] *cos-funpow-bound*[*OF assms*]
 by *auto*
 then have $(\lambda n. (\cos \wedge \wedge n) y0 - dotted) \longrightarrow 0$
 by (*rule tendsto-rabs-zero-cancel*)
 then show ?thesis
 using *Lim-null* by *blast*
 qed

theorem *cos-iter-tendsto-dotted*:

fixes $x0 :: real$
 shows $(\lambda n. (\cos \wedge \wedge n) x0) \longrightarrow dotted$
proof –
 have $|cos x0| \leq 1$ by (*simp add: abs-cos-le-one*)
 from *cos-iter-tendsto-unit*[*OF this*]
 have $(\lambda n. (\cos \wedge \wedge n) (cos x0)) \longrightarrow dotted$.
 moreover have $\bigwedge n. (\cos \wedge \wedge n) (cos x0) = (\cos \wedge \wedge Suc n) x0$
 by (*simp add: funpow-swap1*)
 ultimately have $(\lambda n. (\cos \wedge \wedge Suc n) x0) \longrightarrow dotted$ by *simp*
 then show ?thesis
 using *filterlim-sequentially-Suc* by *blast*
 qed

1.5 A trigonometric identity

Since $\cos(dotted) = dotted$ and $dotted \in (0, 1)$, the Pythagorean identity gives
 $\sin(dotted) = \sqrt{1 - dotted^2}$.

lemma *sin-dotted*: $\sin dotted = \text{sqrt } (1 - dotted^2)$

proof –
 have $0 < dotted < pi$ using *facts pi-gt3* by *auto*
 then have *pos*: $0 < \sin dotted$ by (*rule sin-gt-zero*)
 have $(\sin dotted)^2 = 1 - (\cos dotted)^2$
 using *sin-cos-squared-add*[*of dotted*] by (*simp add: algebra-simps*)
 also have $\dots = 1 - dotted^2$
 using *facts(3)* by *simp*
 finally have $(\sin dotted)^2 = 1 - dotted^2$.

```

then show ?thesis
  using pos by (simp add: real-sqrt-unique)
qed

```

1.6 Transcendence

By the Hermite–Lindemann–Weierstraß theorem, $\cos z$ is transcendental for every nonzero algebraic z . If the Dottie number were algebraic, then $\cos(\text{dottie}) = \text{dottie}$ would be both algebraic and transcendental.

theorem *dottie-transcendental*: \neg *algebraic dottie*

proof

assume *alg: algebraic dottie*

then have \neg *algebraic (cos (complex-of-real dottie))*

using facts *transcendental-cos* **by** *auto*

moreover have *cos (complex-of-real dottie) = complex-of-real dottie*

using facts **by** (*simp add: cos-of-real*)

ultimately show *False* **using** *alg* **by** *simp*

qed

end

We make key facts available outside the locale

lemmas *dottie-fp* = *Dottie.facts(3)*

lemmas *dottie-bounds* = *Dottie.lb Dottie.ub*

lemmas *dottie-attractor* = *Dottie.cos-iter-tendsto-dottie*

lemmas *dottie-sin* = *Dottie.sin-dottie*

lemmas *dottie-transcendental* = *Dottie.dottie-transcendental*

end